This listing based on *10classSoundV4.bas*
It is for the internet description of Picaxe steam sound when made into pdf
****

```
'Steam sound generator with analogue speed input driver.
'NEW BASED ON 20 CLASS TO GIVE SHAPED CHUFF. USES VARYING VOLTAGE ON PIN 2
'TO CONTROL VOLUME. FEEDS TO A CAPACITOR WITH 40MS TC TO SHAPE CHUFF.
'installed in 2-8-2 tank loco 9

'*********Uses 08M type  at 4 MHz ***************************

'version V4     installed  23/2/13              229 bytes


'-----------------------------------------------------------------
'uses voltage across FET for determining speed, so have to subtract reading from
fullscale.(L10:)

'the SOUND command only operates on periods of sound of 12 ms

'Volume is controlled by a dc voltage from PWM on pin2.
'When pin2 is HIGH, the volume is
'set purely by the ext cct but I'm not sure how!!  gives about 50mVpp to spkr.
'reducing the PWM voltage increases volume. minVol gives a slightly louder than
pin2 HIGH
'midVol is used for loudness when slowing. maxVol is loud chuff.
'when set at 650 OP should be (4.7k/26.7k)*(1024-650)/1024*5*20=6Vpp measured 7Vpp!
'when set at 830, OP should be (4.7/26.7)*(1024-830)/1024*5*20=3.3Vpp, meas 3.5Vpp
'when set at 970, OP should be (4.7/26.7)*(1024-970)/1024*100=1Vpp  yes!

Checks battery voltage on start up and 'whistles' if too low.
Every 4th chuff has a longer chuff to give a 'beat'

'pin 0 is the sound output
'Pin1 is speed volts input
'pin2 is varying volume control voltage
'pin3 not used
'pin4 not used
'-------------------------------------------------------------------------
      symbol FETvolts=w0
      symbol speed=b2         'input voltage 1-160 count
      symbol oldspeed=b3      'previous speed
      symbol speed2=b4        'oldspeed+10
      symbol ONtime=b5        'length of hiss sound in 12ms periods
      symbol OFFtime=w3       'quiet time between hiss in ms
      symbol volume=w4        '1=loud, 0=muted
      symbol battV=w5
      symbol counter=b12      'counts chuffs 1-4 for 'beat'

      symbol offVol=970       'by experiment
      symbol midVol=830
      symbol maxVol=650
'-------------------------------------------------------------------------
       HIGH 2                 'min volume
      PAUSE 600        'to charge volume control cap to stop loud chuff at start

        OLDspeed=0

       readADC10 1,battV
      if battV<900 then whistle    'low battery volts

start:   counter=0

NextRead:
```

```
         counter=counter+1
         if counter>4 then start
         if pin3=1 then whistle

read2:  readADC10 1,FETvolts             'for 08M type   1023=maxV=stopped

        if battV>FETvolts then L10
        battV=FETvolts

L10:    speed=battV+40-FETvolts/64*10          'adjust for max speed=160, min=5

        if speed <11 then stopped
        if oldspeed=0 and speed>10 then accel
L30:    speed2=oldspeed+10
L35:    if speed>=speed2 then accel
L40:    if speed<oldspeed then slowing
        goto times2


slowing:                   'quiet chuffs
        oldspeed=speed+10           '+10 necessary
        volume=midVol               'mute the volume
        if speed<50 then Vslow      'as approach stop, just hiss
        goto times2


accel:  oldspeed=speed-15
        volume=maxVol                     'mute off

      'now do the chuffing !!!!!
times2:
        if speed<10 then stopped
        onTime=200/speed+4          '12ms counts
        offTime=600/speed-1         '12 ms
        if counter<4 or volume=midvol then chuffer
        onTime=ontime*5/4           'give it a 'beat'when accelerating

        'low 0          'found that debug only works when pin0 set to zero!!!!!
        'debug

chuffer:
      'this is the chuff, with 47ms TC external cct
      PWMOUT 2,255, volume          '1=loud,  1023=off
      SOUND 0, (255, ontime)

      'now turn off the volume and let it decay with ext cct
      PWMOUT 2,255,offvol
      SOUND 0, (255, offtime) 'sounds much better than having silence during offtime

      goto nextread



'+++++++++++++++++++++++++++++++++++++++++++++

stopped:
      if oldspeed>10 then justStopped
      pwmout 2, off           'min volume
      high 2
      SOUND 0, (255, 50)
      goto nextread

justStopped:
      PAUSE 500
```

```
      PWMOUT 2,255, midvol          '1=loud,  1023=off
      SOUND 0, (255, 100)           'exhaust brakes sound
      PWMOUT 2,255, offvol
      SOUND 0,(255,100)
      oldspeed=0
      goto nextread


Vslow:    'come here as approach stop and want to just hiss.

      PWMOUT 2,255, offvol          '1=loud,  1023=off
      SOUND 0, (255, 50)
      goto nextread

Whistle:                'also low volts at startup
      PWMOUT 2,255,maxVol
      SOUND 0,(112,100)
      PWMOUT 2,255,offvol
      pause 100               'to stop single loud chuff after whistle  why does
it???
'***should change this pause to SOUND 0,(111,8)

      goto nextread


#####################################################################

Use with this circuit:   (steam cct4.jpg)
```